

Project Euler 9. Special Pythagorean Triplet

hiragn

2024 年 12 月 23 日

1. 問題の概要

$a^2 + b^2 = c^2$, $a + b + c = 1000$, $a < b < c$ をみたすピタゴラス数がただ一つ存在する。これらの積 abc を求めよ。

<https://projecteuler.net/problem=9>

2. Solve で整数解を求める

与えられた方程式を Solve で解くと (200, 375, 425) という解が見つかります。これらの積が答えです。

```
1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
3   eqn = {x^2 + y^2 == z^2, x + y + z == 1000, z > y > x > 0};
4   sol = First@Values@Solve[eqn, {x, y, z}, Integers];
5   ans = {Times @@ sol, sol}]
6
7 Out[]:= {0.0463766, {31875000, {200, 375, 425}}}
```

3. 整数の分割

IntegerPartitions[1000, 3] で 1000 を 3 つの自然数の和に分割して、ピタゴラス数を探します。この関数の返り値は降順なので抽出条件は $x^2 + y^2 = z^2$ ではなく、 $x^2 = y^2 + z^2$ になります。

```
1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
```

```

3 cond[{x_, y_, z_}] := x^2 == y^2 + z^2;
4 ans = Times @@@ Select[IntegerPartitions[1000, {3}], cond]]
5
6 Out[] = {0.10233, {31875000}}
```

4. 二重ループ

a, b についてループさせます。 $c = 1000 - a - b$ が $a^2 + b^2 = c^2$ をみたしたら abc を返して終了です。

```

1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   n = 1000;
4   ans = Catch[
5     For[a = 1, a < n, a++,
6       For[b = a + 1, c = n - a - b; b < c, b++,
7         If[a^2 + b^2 == c^2, Throw[a*b*c]]]]]]
8
9 Out[] = {0.102165, 31875000}
```

5. ピタゴラス数の一般形

ピタゴラス数は自然数 k, m, n ($m > n$) を用いて次のようにあらわせます。 m と n は互いに素で偶奇が異なる数です。

$$(a, b, c) = (k(m^2 - n^2), 2kmn, k(m^2 + n^2))$$

$a + b + c = 1000$ は $km(m + n) = 500$ と書き換えられます。500 の約数は 12 個しかないのでこの解法は速いです。

1. 500 の約数 k を選ぶ
2. $500/k$ の約数 m を選ぶと $n = 500/k - m$ と決まる
3. m, n の条件をチェック。パスしたら abc に当たる値を返して終了

```

1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   calc[m_, n_, k_] := k^3 (m^2 - n^2)*(2*m*n)*(m^2 + n^2);
4   cond[m_, n_] := CoprimeQ[m, n] && OddQ[m - n] && m > n;
5   ans = Catch[Do[
6     n = Quotient[500, k*m] - m;
```

```
7      If[cond[m, n], Throw@calc[m, n, k]],  
8      {k, Divisors@500}, {m, Most@Divisors@Quotient[500, k]}]]]  
9  
10 Out[] = {0.0000791069, 31875000}
```
