

Project Euler 81. Path Sum: Two Ways

hiragn

2024 年 12 月 26 日

1. 問題の概要

下の 5 次の正方行列で左上のセルから出発して右下のセルで終わる経路を考える。下方向と右方向にのみ移動できるものとする。

経路上の数の総和が最小となるパスは赤字で示されたもので、その値は 2427 である。

$$\begin{pmatrix} 131 & 673 & 234 & 103 & 18 \\ 201 & 96 & 342 & 965 & 150 \\ 630 & 803 & 746 & 422 & 111 \\ 537 & 699 & 497 & 121 & 956 \\ 805 & 732 & 524 & 37 & 331 \end{pmatrix}$$

テキストファイル 0081_matrix.txt^a には 80 × 80 の行列が書かれている。左上のセルから出発して、右方向と下方向にのみ移動して右下のセルで終わる経路を考える。経路上の数の総和の最小値を求めよ。

<https://projecteuler.net/problem=5>

^a https://projecteuler.net/project/resources/0081_matrix.txt

2. 動的計画法

動的計画法 (dp) を使います。

$$dp[i, j] = \text{mat}[i, j] + \text{Min}[dp[i, j - 1], dp[i - 1, j]]$$

```

1 In[] := Clear["Global`*"];
2 mat = Import["0081_matrix.txt", "CSV"];
3 RepeatedTiming[
4   dp[1, 1] = mat[[1, 1]];
5   dp[1, j_ /; j >= 2] := mat[[1, j]] + dp[1, j - 1];
6   dp[i_, 1] /; i >= 2 := mat[[i, 1]] + dp[i - 1, 1];
7   dp[i_, j_] :=
8     dp[i, j] = mat[[i, j]] + Min[dp[i, j - 1], dp[i - 1, j]];
9   ans = dp[80, 80]]
10
11 Out[] = {6.31727*10^-6, 427337}
```

3. グラフ理論

ちょっと大げさなのですが、次の問題、次の次の問題のことも考えてグラフ理論でも解いてみました。

出発点とゴールの外側にそれぞれ st, gl という点を追加し、セルに書かれた数字を辺の重みとしてもつ有向グラフを作ります。このグラフ上で st と gl の最短距離が答えです。例として与えられた 5×5 の表でやってみましょう。

```

1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   mat = {{131, 673, 234, 103, 18},
4     {201, 96, 342, 965, 150},
5     {630, 803, 746, 422, 111},
6     {537, 699, 497, 121, 956},
7     {805, 732, 524, 37, 331}};
8   calc[lst_] := Module[{m, n, v},
9     {m, n} = Dimensions@lst;
10    Graph[Flatten[Join[{st -> v[1, 1]},
11      Table[v[i, j] -> v[i, j + 1], {i, 1, m}, {j, 1, n - 1}], (* right *)
12      Table[v[i, j] -> v[i + 1, j], {i, 1, m - 1}, {j, 1, n}], (* down *)
13      {v[m, n] -> gl}], (* goal *)],
14    EdgeWeight -> Flatten[Join[{mat[[1, 1]]},
15      Table[mat[[i, j + 1]], {i, 1, m}, {j, 1, n - 1}],
```

```

16      Table[mat[[i + 1, j]], {i, 1, m - 1}, {j, 1, n}], {0}]]]]];
17  ans = Floor@GraphDistance[calc@mat, st, gl]]
18
19 Out[] = {0.0000921772, 2427}

```

同じことを与えられた 80×80 の経路でやりました。

```

1 In[] := Clear["Global`*"];
2 mat = Import["0081_matrix.txt", "CSV"];
3 RepeatedTiming[
4   calc[lst_] := Module[{m, n, v},
5     {m, n} = Dimensions@lst;
6     Graph[Flatten[Join[{st -> v[1, 1]},
7       Table[v[i, j] -> v[i, j + 1], {i, 1, m}, {j, 1, n - 1}],
8       Table[v[i, j] -> v[i + 1, j], {i, 1, m - 1}, {j, 1, n}],
9       {v[m, n] -> gl}]],
10    EdgeWeight -> Flatten[Join[{mat[[1, 1]]},
11      Table[mat[[i, j + 1]], {i, 1, m}, {j, 1, n - 1}],
12      Table[mat[[i + 1, j]], {i, 1, m - 1}, {j, 1, n}], {0}]]]]];
13  ans = Floor@GraphDistance[calc@mat, st, gl]]
14
15 Out[] = {0.0188522, 427337}

```
