

Project Euler 31. Coin Sums

hiragn

2024 年 12 月 23 日

1. 問題の概要

イギリスでは硬貨はポンド \pounds とペンス p があり、一般的に流通している硬貨は以下の 8 種類である。

1p, 2p, 5p, 10p, 20p, 50p, $\pounds 1$ (100p), $\pounds 2$ (200p)

これらの硬貨を使って $\pounds 2$ を作る方法は何通りあるか？

<https://projecteuler.net/problem=31>

要するに 1, 2, 5, 10, 20, 50, 100, 200 を使って 200 を作る問題です。

2. 整数の分割

IntegerPartitions を使って 200 を分割します。

IntegerPartitions[a,b,list] とすれば list に含まれる数を最大 b 個使って a を表す方法が返ってきます。この関数で 200 を分割しました。

```
1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
3   coins = {1, 2, 5, 10, 20, 50, 100, 200};
4   ans = Length@IntegerPartitions[200, All, coins]]
5
6 Out[] = {0.0420703, 73682}
```

3. べき級数展開

べき級数を使います。たとえば 1, 2, 5 を使って 10 を作る方法は

$$(1 + x + x^2 + \cdots)(1 + x^2 + x^4 + \cdots)(1 + x^5 + x^{10} + \cdots)$$

を展開したときの x^{10} の係数と同じ数だけあります。級数を整理すると次のようになり、これの 10 次の係数を求めればいいわけです。Series と Coefficient が使えます。

$$\frac{1}{(1-x)(1-x^2)(1-x^5)}$$

```
In[]:= f[x_] := 1/((1 - x) (1 - x^2) (1 - x^5));
Coefficient[Series[f@x, {x, 0, 10}], x, 10]
Out[]= 10
```

同じ方法で 200 を作ります。

```
1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
3   f[x_] := 1/((1 - x) (1 - x^2) (1 - x^5) (1 - x^10)
4     (1 - x^20) (1 - x^50) (1 - x^100) (1 - x^200));
5   ans = Coefficient[Series[f@x, {x, 0, 200}], x, 200]]
6
7 Out[]= {0.000168222, 73682}
```

4. 動的計画法

8 種類のコインを i 種類目まで見て j ペンス作る方法が $dp[i, j]$ 通りあるとします。これは次のような場合に対応します。

- $i - 1$ 種類目までで j ペンスになっていて、 i 種類目のコインは使わない
- i 種類目まで見ていて、 i 種類目のコインを (もう 1 枚) 使う

漸化式は次のようになります。

$$dp[i, j] = dp[i - 1, j] + dp[i, j - \text{coins}[i]]$$

初期条件や自明なケースをおさえてコードを書きます。これが一番速いです。

- $dp[i, 0] = 1$
- $i \leq 0$ または $j < 0$ のときは $dp[i, j] = 0$

```
1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   coins = {1, 2, 5, 10, 20, 50, 100, 200};
4   dp[i_, 0] := dp[i, 0] = 1;
5   dp[i_ /; i <= 0, j_] := dp[i, j] = 0;
6   dp[i_, j_ /; j < 0] := dp[i, j] = 0;
7   dp[i_, j_] := dp[i, j] = dp[i - 1, j] + dp[i, j - coins[[i]]];
8   ans = dp[8, 200]]
9
10 Out[] = {7.32598*10^-6, 73682}
```
