

Project Euler 93. Arithmetic Expressions

hiragn

2024 年 12 月 24 日

1. 問題の概要

集合 $\{1, 2, 3, 4\}$ の各数をちょうど 1 回ずつ使い、四則演算（加減乗除）と括弧を使うと異なる正の整数を作ることができる。

$$8 = (4 \times (1 + 3)) / 2$$

$$14 = 4 \times (3 + 1/2)$$

$$19 = 4 \times (2 + 3) - 1$$

$$36 = 3 \times 4 \times (2 + 1)$$

$12 + 34$ のように数字をつなげることは許されないことに注意しよう。

集合 $\{1, 2, 3, 4\}$ から 36 を最大とする 31 個の異なる数が得られる。最初の表現できない数に会うまで 1 から 28 の各数を得ることができる。

1 からはじまる最も長い連続整数列を作れる 4 つの数 a, b, c, d ($a < b < c < d$) を求めて $abcd$ の形で答えよ。

<https://projecteuler.net/problem=93>

2. 解法

「式のどこに括弧をはさむか」「どの演算子を使うか」に分けて考えます。

「括弧の対応→カタラン数→トーナメント表」と連想してググりました。4 人のトーナメント表の書き方は 5 通りしかなく、括弧の位置は次の 5 つを考えれば十分です。

- $h(f(x, y), g(z, w))$
- $h(g(f(x, y), z), w)$
- $h(f(x, g(y, z)), w)$
- $h(x, g(y, f(z, w)))$

$$\bullet h(x, g(f(y, z), w))$$

f, g, h は $+$, $-$, \times , \div のどれかです。計算式は全部で $5 \cdot 4^3 = 320$ 個あります。

そして問題文中の例からわかるように a, b, c, d は小さいものから順に代入するとは限りません。これらの代入法は $9 \cdot 8 \cdot 7 \cdot 6 = 3024$ 通りです。

コードはこうなりました。

```

1 In[] := ClearAll["Global`*"];
2 RepeatedTiming[
3   ops = {Plus, Subtract, Times, Divide};
4   f[{x_, y_, z_, w_}] := Table[
5     {r[p[x, y], q[z, w]],
6     r[q[p[x, y], z], w],
7     r[p[x, q[y, z]], w],
8     r[x, q[y, p[z, w]]],
9     r[x, q[p[y, z], w]]},
10    {p, ops}, {q, ops}, {r, ops}];
11 g[lst_] := Module[{tlst1, tlst2, tlst3},
12   tlst1 = DeleteDuplicates@Flatten[f /@ Permutations@lst];
13   tlst2 = Sort@Select[Select[tlst1, IntegerQ], # > 0 &];
14   If[First@tlst2 != 1, Return@0];
15   tlst3 = First@SplitBy[Differences@tlst2, # != 1 &];
16   Length@tlst3 + 1];
17 lst = Subsets[Range@9, {4}];
18 ans = FromDigits@First@MaximalBy[lst, g] // Quiet
19
20 Out[] = {0.479495, 1258}

```

(a, b, c, d) を計算式に代入すると 0 による除算がおこったり、負の数や分数になったりします。正の整数だけ抽出してソートしたのがコード中の `tlst2` です。

この先頭が 1 でない場合は 1 を作れないので、条件をみません。Return 0 します。

`tlst2` の先頭が 1 のときはどの数まで表せるか調べます。

1. `Differences` で階差数列を作る。連続する整数なら値は 1 になる
2. 1 以外の値が出たらそこから先は捨てる
3. 残ったリストの長さを求める。この長さプラス 1 の数まで表せる

最も長い連続整数列を作れるのは $\{1, 2, 5, 8\}$ でした。1 から 51 まで作れます。

1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19,
20, 21, 22, 23, 24, 25, 26, 27, 28, 29, 30, 31, 32, 33, 34, 35, 36,
37, 38, 39, 40, 41, 42, 43, 44, 45, 46, 47, 48, 49, 50, 51, 55, 56,
57, 60, 63, 64, 70, 72, 75, 78, 79, 80, 81, 82, 85, 88, 90, 96, 120