

Project Euler 42. Coded Triangle Numbers

hiragn

2024 年 12 月 23 日

1. 問題の概要

三角数は $t(n) = n(n+1)/2$ で与えられ、最初の 10 項は次のようになる。

1, 3, 6, 10, 15, 21, 28, 36, 45, 55, ...

単語中のアルファベットを数値に変換した後に和をとり、この和を「単語の値」ということにする。たとえば SKY の値は 55 である。

$$19 + 11 + 25 = 55 = t(10)$$

単語の値が三角数であるとき、その単語を三角語ということにする。テキストファイル^a中に約 2000 語の英単語が記されている。三角語はいくつあるか？

<https://projecteuler.net/problem=42>

^a https://projecteuler.net/project/resources/0042_words.txt

2. 単語の値リストと三角数リストの比較

単語の値リストと三角数リストの共通項を数えて解きました。

```
1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   calc[s_] := Total@LetterNumber@s;
4   wList = calc /@ Flatten[Import["0042_words.txt", "CSV"]];
5   kMax = NestWhile[# + 1 &, 1, PolygonalNumber@# < Max@wList &];
6   pList = Table[PolygonalNumber@k, {k, kMax}];
7   ans = Length@Select[wList, MemberQ[pList, #] &]]
8 Out[] = {0.139008, 162}
```

`calc(s)` は単語の値を求める関数で, `wList` は与えられた単語に対応する値のリストです。
`kMax` で三角数を第何項まで取ればいいのか調べています。単語の値の最大値は 192 だったので $t(20) = 210$ までで十分でした。

`pList` は三角数のリストです。`MemberQ` で `pList` と `wList` の共通項を数えています。

3. 三角数の判定

三角数の判定を他の方法でやってみました。 $n(n+1)/2 = a$ を解くと

$$n = \frac{\sqrt{1+8a}-1}{2}$$

a が三角数になる条件は $1+8a$ が奇数の平方数であることです。奇数であることは明らかなので平方数かどうか調べます。計算時間はほぼ同じでした。

```

1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
3   calc[s_] := Total@LetterNumber@s;
4   lst = calc /@ Flatten[Import["0042_words.txt", "CSV"]];
5   squareQ[n_] := AllTrue[FactorInteger[1 + 8 n], EvenQ[#[[2]]] &];
6   ans = Length@Select[lst, squareQ]]
7
8 Out[]= {0.140557, 162}
```
