

Project Euler 61. Cyclical Figurate Numbers

hiragn

2024 年 12 月 26 日

1. 問題の概要

三角数, 四角数, 五角数, 六角数, 七角数, 八角数は以下の式で与えられる。

$$\text{三角数} \quad P_{3,n} = n(n+1)/2 \quad 1, 3, 6, 10, 15, \dots$$

$$\text{四角数} \quad P_{4,n} = n^2 \quad 1, 4, 9, 16, 25, \dots$$

$$\text{五角数} \quad P_{5,n} = n(3n-1)/2 \quad 1, 5, 12, 22, 35, \dots$$

$$\text{六角数} \quad P_{6,n} = n(2n-1) \quad 1, 6, 15, 28, 45, \dots$$

$$\text{七角数} \quad P_{7,n} = n(5n-3)/2 \quad 1, 7, 18, 34, 55, \dots$$

$$\text{八角数} \quad P_{8,n} = n(3n-2) \quad 1, 8, 21, 40, 65, \dots$$

3 つの 4 桁の数の順番つき集合 (8128, 2882, 8281) は面白い性質をもつ。

- この集合は巡回的である。最後の数も含めて各数の後半 2 桁は次の数の前半 2 桁と一致する
- それぞれ多角数である。三角数 ($P[3, 127] = 8128$), 四角数 ($P[4, 91] = 8281$), 五角数 ($P[5, 44] = 2882$) がそれぞれ別の数で集合に含まれている

4 桁の数の組で上の 2 つの性質をもつのはこの組だけである。

三角数, 四角数, 五角数, 六角数, 七角数, 八角数がすべて表れる 6 つの巡回する 4 桁の数からなる唯一の順序集合の和を求めよ。

<https://projecteuler.net/problem=61>

2. 解法

「ある数の後半 2 桁が次の数の前半 2 桁と一致する」をグラフ理論で処理します。

条件をみたす数 i, j の間に有向辺を張ってグラフを作り、長さ 6 のサイクルを探しました。

サイクルはどこからはじまってもいいので、三角数からはじまって三角数で終わるものとして計算しています。21 行目で 4~8 を並び替えたリストを作り、18 行目で `lst` の前後に `{3}` を Join しているのがこの部分です。

```

1 In[] := Clear["Global`*"];
2 RepeatedTiming[
3   (* 4桁のk角数の集合 *)
4   nums[k_] := nums[k] = Module[{lft, rght},
5     lft = NestWhile[# + 1 &, 1, PolygonalNumber[k, #] < 10^3 &];
6     rght = NestWhile[# + 1 &, lft, PolygonalNumber[k, #] < 10^4 &] - 1;
7     PolygonalNumber[k, #] & /@ Range[lft, rght]];
8
9   (* i角数とj角数をむすぶグラフ *)
10  edge[i_, j_] := edge[i, j] =
11    If[Mod[i, 100] == Quotient[j, 100], DirectedEdge[i, j], Nothing];
12  gr[{i_, j_}] := gr[{i, j}] =
13    Graph@Flatten@Union@Outer[edge, nums@i, nums@j];
14
15  (* サイクルを探す *)
16  solve[lst_] := Module[{found},
17    found = Flatten@FindCycle[GraphUnion @@
18      (gr /@ (Partition[Join[{3}, lst, {3}], 2, 1))), {6}];
19    If[found == {},
20      Nothing, {Total[Total /@ VertexList@found], lst}];
21  ans = First@(solve /@ Permutations@Range[4, 8])]
22
23 Out[] = {1.08736, {28684, {4, 7, 8, 6, 5}}}
```

ちなみに条件をみたすサイクルは

$$1281 \rightarrow 8128 \rightarrow 2882 \rightarrow 8256 \rightarrow 5625 \rightarrow 2512 \rightarrow 1281$$

です。順に 3, 4, 7, 8, 6, 5, 3 角数でした。