Project Euler 86. Cuboid Route

hiragn

2024 年 12 月 27 日

1.　問題の概要

$6 \times 5 \times 3$ の直方体の頂点 S から頂点 F まで直方体の表面を移動する最短ルートの長さは 10 である。



$M \times M \times M$ 以下の直方体で最短ルートの長さが整数であるものの個数を考える。
$M = 99$ のときは 1975 個で，$M = 100$ のときは 2060 個である。$M = 100$ は 2000 個を超える最小の M である。直方体の個数が 100 万個を超える最小の M を求めよ。

https://projecteuler.net/problem=86

2.　固定して数える

3 辺の長さを $a,\, b,\, c\,(a \leqq b \leqq c \leqq m)$ とおきます。
最短ルートの長さは $\sqrt{(a+b)^2+c^2}$ で，これが整数になる条件を考えます。
$c = 1,\, 2,\, \cdots$ と増やしていって，条件をみたす $(a,\, b,\, c)$ の個数がはじめて 100 万を超えたときの c が答えです。以下，m ではなく c で考えます。
$(a,\, b)$ をそのまま相手にするのは面倒なので $a+b=x$ とおいて次の手順で数えます。

1. x^2+c^2 が平方数になる x を抽出する
2. 各 x に対応する $(a,\, b)$ の個数を求めて足しあわせる

$x \to (a,\, b)$ を `Solve` で処理すると約 1 分かかります。

```
1 In[]:= Clear["Global`*"];
2 RepeatedTiming[
3  squareQ[n_] := AllTrue[FactorInteger@n, EvenQ[#[[2]]] &];
4
5  (* x=a+b に対応する(a,b)の個数 *)
6  f[x_, c_] := Module[{a, b, eqn},
7    eqn = {a + b == x, 1 <= a <= b <= c};
8    Length@Values@Solve[eqn, {a, b}, Integers]];
9
10  (* 計算本体 *)
11  target = 10^6;
12  cnt = 0;
13  c = 1;
14  While[cnt <= target,
15   cnt +=
16    Total[f[#, c] & /@ Select[Range[2, 2 c], squareQ[#^2 + c^2] &]];
17   c++]; ans = c - 1]
18
19 Out[]= {58.2719, 1818}
```

3. 格子点を数える

実は上のコード 6 行目の $f[x,c]$ は具体形を求めることができます。

ab 平面で考えると，これは線分 $a+b=x$ $(1 \leqq a \leqq b \leqq c)$ 上の格子点の個数です。図（省略）を描いて数えるとこうなります。

- $x \leqq c$ のとき $f[x,c] = \texttt{floor}(x/2)$
- $x > c$ のとき $f[x,c] = \texttt{floor}(x/2) - (x-c) + 1$

この解法だと 10 秒かかりません。

```
In[]:= Clear["Global`*"];
RepeatedTiming[
 squareQ[n_] := AllTrue[FactorInteger@n, EvenQ[#[[2]]] &];

 (* x=a+b に対応する(a,b)の個数 *)
 f[x_, c_] := Quotient[x, 2] /; x <= c;
 f[x_, c_] := Quotient[x, 2] - (x - c) + 1 /; x > c;

 (* 計算本体 *)
 target = 10^6;
```

```
cnt = 0;
c = 1;
While[cnt <= target,
 cnt +=
  Total[f[#, c] & /@ Select[Range[2, 2 c], squareQ[#^2 + c^2] &]];
 c++]; ans = c - 1]

Out[]= {8.40147, 1818}
```